

This module verifies the *SeekLowerBound* algorithm in the *go-immutable-radix* Go library (<https://github.com/hashicorp/go-immutable-radix>).

MODULE *RadixSeekLowerBound*

EXTENDS *FiniteSets*, *Integers*, *Sequences*, *TLC*

Set of characters to use for the alphabet of generated strings.

CONSTANT *Alphabet*

CmpOp is the comparison operator for ordered iteration. This should be TRUE if the first value is less than the second value. This is called on a single element of a sequence.

CONSTANT *CmpOp*($_$, $_$)

Length of input strings generated

CONSTANT *MinLength*, *MaxLength*

ASSUME

$\wedge \{MinLength, MaxLength\} \subseteq Nat$

$\wedge MinLength \leq MaxLength$

Number of unique elements to construct the radix tree with. This is a set of numbers so you can test with inputs of multiple sizes.

CONSTANT *ElementCounts*

ASSUME *ElementCounts* $\subseteq Nat$

INSTANCE *RadixTrees*

INSTANCE *RadixIterator*

Inputs is the set of input strings valid for the tree.

Inputs $\triangleq \text{UNION } \{[1 \dots n \rightarrow Alphabet] : n \in MinLength \dots MaxLength\}$

InputSets is the full set of possible inputs we can send to the radix tree.

InputSets $\triangleq \{T \in \text{SUBSET } Inputs : Cardinality(T) \in ElementCounts\}$

TRUE iff the sequence *s* contains no duplicates. Copied from *CommunityModules*.

isInjective(*s*) $\triangleq \forall i, j \in \text{DOMAIN } s : (s[i] = s[j]) \Rightarrow (i = j)$

Converts a set to a sequence that contains all the elements of *S* exactly once.

Copied from *CommunityModules*.

setToSeq(*S*) $\triangleq \text{CHOOSE } f \in [1 \dots Cardinality(S) \rightarrow S] : isInjective(f)$

bytes.Compare in Go

RECURSIVE *GoBytesCompare*($_$, $_$)

GoBytesCompare(*X*, *Y*) \triangleq

CASE *X* = *Y* $\rightarrow 0$

□ *Len*(*X*) = 0 $\rightarrow -1$

□ *Len*(*Y*) = 0 $\rightarrow 1$

```

□ OTHER →
  IF  $X[1] = Y[1]$ 
    THEN  $GoBytesCompare(Tail(X), Tail(Y))$ 
    ELSE IF  $CmpOp(X[1], Y[1])$  THEN  $-1$  ELSE  $1$ 

```

$CmpSeq$ compares two full inputs whereas $CmpOp$ compares only a single element of the alphabet.

$CmpSeq(X, Y) \triangleq GoBytesCompare(X, Y) \leq 0$

$CmpGte$ checks if $X \geq Y$

$CmpGte(X, Y) \triangleq X = Y \vee \neg CmpOp(X, Y)$

Sorted edge labels based on $CmpOp$.

$SortedEdgeLabels(Node) \triangleq SortSeq(setToSeq(DOMAIN Node.Edges), CmpOp)$

Returns the index of the first element that is greater than or equal to to the search label.

```

GetLowerBoundEdgeIndex(Node, Label)  $\triangleq$ 
  IF  $\neg \exists e \in DOMAIN Node.Edges : e = Label \vee \neg CmpOp(e, Label)$  THEN 0
    if there is no lower bound edge, return 0
  ELSE LET
     $e \triangleq SortedEdgeLabels(Node)$ 
    sorted edges
  IN CHOOSE  $idx \in 1 .. Len(e) :$  find the index
     $\wedge CmpGte(e[idx], Label)$   $\geq$  to our search label
     $\wedge \vee idx = 1$  and its the first element that is gte
     $\vee CmpOp(e[idx - 1], Label)$ 

```

The expected value is the sorted set of all inputs where the element is greater than or equal to the given key.

EXPLANATION:

1. We convert the input set to a sequence
2. Sort the input sequence, this is all inputs sorted now.
3. Select the subset of the input sequence where it satisfies our comparison.

The sequence now only has elements greater than or equal to our key

$Expected(input, key) \triangleq$
 $SelectSeq(SortSeq(setToSeq(input), CmpSeq), LAMBDA elem : CmpSeq(key, elem))$

--algorithm *seek_lower_bound*

variables

```

iterStack =  $\langle \rangle$ ,
input  $\in InputSets$ ,
key  $\in Inputs$ ,
root =  $RadixTree(input)$ ,

```

```

node = {},
search = {},
result = {},
prefixCmp = "UNSET" ;

findMin as implemented in Go
procedure findMin()begin
FindMin:
  while Len(node.Value) = 0 do
    with
      labels = SortedEdgeLabels(node),
      edges = [n ∈ 1 .. Len(labels) ↦ node.Edges[labels[n]]]
    do
      if Len(edges) > 1 then
        iterStack := iterStack ◦ SubSeq(edges, 2, Len(edges)) ;
      end if ;

      if Len(edges) > 0 then
        recurse again
        node := edges[1] ;
      else
        shouldn't be possible
      return ;
      end if ;
    end with ;
  end while ;

  iterStack := iterStack ◦ ⟨node⟩ ;
  return ;
end procedure ;

```

This entire algorithm is almost 1:1 translated where possible from the actual implementation in *iter.go*. That's the point: we're trying to verify our algorithm is correct for all inputs.

Source: <https://github.com/hashicorp/go-immutable-radix/blob/f63f49c0b598a5ead21c5015fb4d08fe7e3c21ea/iter.go> ≠ L77

```

begin
  I could've just set these variables in the initializer above but
  to better closely match the algorithm, I reset them here.
Begin:
  iterStack := ⟨⟩ ;
  node := root ;
  search := key ;

```

```

Seek:
  while TRUE do

```

```

if  $Len(node.Prefix) < Len(search)$  then
   $prefixCmp := GoBytesCompare(node.Prefix, SubSeq(search, 1, Len(node.Prefix)))$ ;
else
   $prefixCmp := GoBytesCompare(node.Prefix, search)$ ;
end if ;

if  $prefixCmp < 0$  then
  goto Result ;
elsif  $prefixCmp > 0$  then
  call findMin() ;
  goto Result ;
end if ;

Search:
if  $Len(node.Value) > 0 \wedge node.Value = key$  then
   $iterStack := iterStack \circ \langle node \rangle$  ;
  goto Result ;
end if ;

Consume:
 $search := SubSeq(search, Len(node.Prefix) + 1, Len(search))$  ;

if  $Len(search) = 0$  then
  call findMin() ;
  goto Result ;
end if ;

NextEdge:
with
   $idx = GetLowerBoundEdgeIndex(node, search[1])$ ,
   $labels = SortedEdgeLabels(node)$ ,
   $edges = [n \in 1 \dots Len(labels) \mapsto node.Edges[labels[n]]]$ 
do
  if  $idx = 0$  then
    goto Result ;
  else
    if  $idx + 1 \leq Len(edges)$  then
       $iterStack := iterStack \circ SubSeq(edges, idx + 1, Len(edges))$  ;
    end if ;

     $node := edges[idx]$  ;
  end if ;
end with ;

end while ;

Result:
 $result := Iterate(iterStack)$  ;

```

```

CheckResult:
  assert result = Expected(input, key);
end algorithm ;

```

!!! NOTE !!! The rest of the file is auto-generated based on the *PlusCal* above. For those who are reading this to learn TLA+/*PlusCal*, you can stop reading here.

```

BEGIN TRANSLATION (chksum(pcal) = "7f5569db" ∧ chksum(tla) = "177f60c")
VARIABLES iterStack, input, key, root, node, search, result, prefixCmp, pc,
          stack

```

```

vars ≜ ⟨iterStack, input, key, root, node, search, result, prefixCmp, pc,
        stack⟩

```

```

Init ≜ Global variables
      ∧ iterStack = ⟨⟩
      ∧ input ∈ InputSets
      ∧ key ∈ Inputs
      ∧ root = RadixTree(input)
      ∧ node = {}
      ∧ search = {}
      ∧ result = {}
      ∧ prefixCmp = "UNSET"
      ∧ stack = ⟨⟩
      ∧ pc = "Begin"

```

```

FindMin ≜ ∧ pc = "FindMin"
          ∧ IF Len(node.Value) = 0
            THEN ∧ LET labels ≜ SortedEdgeLabels(node) IN
                  LET edges ≜ [n ∈ 1 .. Len(labels) ↦ node.Edges[labels[n]]] IN
                    ∧ IF Len(edges) > 1
                      THEN ∧ iterStack' = iterStack ∘ SubSeq(edges, 2, Len(edges))
                          ELSE ∧ TRUE
                          ∧ UNCHANGED iterStack
                    ∧ IF Len(edges) > 0
                      THEN ∧ node' = edges[1]
                          ∧ pc' = "FindMin"
                          ∧ stack' = stack
                      ELSE ∧ pc' = Head(stack).pc
                          ∧ stack' = Tail(stack)
                          ∧ node' = node
                    ELSE ∧ iterStack' = iterStack ∘ ⟨node⟩
                        ∧ pc' = Head(stack).pc

```

$$\begin{aligned}
& \wedge stack' = Tail(stack) \\
& \wedge node' = node \\
& \wedge \text{UNCHANGED } \langle input, key, root, search, result, prefixCmp \rangle \\
findMin & \triangleq FindMin \\
Begin & \triangleq \wedge pc = \text{"Begin"} \\
& \wedge iterStack' = \langle \rangle \\
& \wedge node' = root \\
& \wedge search' = key \\
& \wedge pc' = \text{"Seek"} \\
& \wedge \text{UNCHANGED } \langle input, key, root, result, prefixCmp, stack \rangle \\
Seek & \triangleq \wedge pc = \text{"Seek"} \\
& \wedge \text{IF } Len(node.Prefix) < Len(search) \\
& \quad \text{THEN } \wedge prefixCmp' = GoBytesCompare(node.Prefix, SubSeq(search, 1, Len(node.Prefix))) \\
& \quad \text{ELSE } \wedge prefixCmp' = GoBytesCompare(node.Prefix, search) \\
& \wedge \text{IF } prefixCmp' < 0 \\
& \quad \text{THEN } \wedge pc' = \text{"Result"} \\
& \quad \quad \wedge stack' = stack \\
& \quad \text{ELSE } \wedge \text{IF } prefixCmp' > 0 \\
& \quad \quad \quad \text{THEN } \wedge stack' = \langle [procedure \mapsto \text{"findMin"}, \\
& \quad \quad \quad \quad pc \mapsto \text{"Result"}] \rangle \\
& \quad \quad \quad \quad \quad \circ stack \\
& \quad \quad \quad \wedge pc' = \text{"FindMin"} \\
& \quad \quad \text{ELSE } \wedge pc' = \text{"Search"} \\
& \quad \quad \quad \wedge stack' = stack \\
& \wedge \text{UNCHANGED } \langle iterStack, input, key, root, node, search, result \rangle \\
Search & \triangleq \wedge pc = \text{"Search"} \\
& \wedge \text{IF } Len(node.Value) > 0 \wedge node.Value = key \\
& \quad \text{THEN } \wedge iterStack' = iterStack \circ \langle node \rangle \\
& \quad \quad \wedge pc' = \text{"Result"} \\
& \quad \text{ELSE } \wedge pc' = \text{"Consume"} \\
& \quad \quad \wedge \text{UNCHANGED } iterStack \\
& \wedge \text{UNCHANGED } \langle input, key, root, node, search, result, prefixCmp, stack \rangle \\
Consume & \triangleq \wedge pc = \text{"Consume"} \\
& \wedge search' = SubSeq(search, Len(node.Prefix) + 1, Len(search)) \\
& \wedge \text{IF } Len(search') = 0 \\
& \quad \text{THEN } \wedge stack' = \langle [procedure \mapsto \text{"findMin"}, \\
& \quad \quad \quad pc \mapsto \text{"Result"}] \rangle \\
& \quad \quad \quad \quad \circ stack \\
& \quad \quad \wedge pc' = \text{"FindMin"} \\
& \quad \text{ELSE } \wedge pc' = \text{"NextEdge"}
\end{aligned}$$

$$\begin{aligned}
& \wedge stack' = stack \\
& \wedge \text{UNCHANGED } \langle iterStack, input, key, root, node, result, \\
& \quad prefixCmp \rangle \\
NextEdge & \triangleq \wedge pc = \text{"NextEdge"} \\
& \wedge \text{LET } idx \triangleq GetLowerBoundEdgeIndex(node, search[1]) \text{ IN} \\
& \quad \text{LET } labels \triangleq SortedEdgeLabels(node) \text{ IN} \\
& \quad \quad \text{LET } edges \triangleq [n \in 1 \dots Len(labels) \mapsto node.Edges[labels[n]]] \text{ IN} \\
& \quad \quad \text{IF } idx = 0 \\
& \quad \quad \quad \text{THEN } \wedge pc' = \text{"Result"} \\
& \quad \quad \quad \quad \wedge \text{UNCHANGED } \langle iterStack, node \rangle \\
& \quad \quad \quad \text{ELSE } \wedge \text{IF } idx + 1 \leq Len(edges) \\
& \quad \quad \quad \quad \quad \text{THEN } \wedge iterStack' = iterStack \circ SubSeq(edges, idx + 1, Len(edges)) \\
& \quad \quad \quad \quad \quad \text{ELSE } \wedge \text{TRUE} \\
& \quad \quad \quad \quad \quad \quad \wedge \text{UNCHANGED } iterStack \\
& \quad \quad \quad \quad \quad \quad \wedge node' = edges[idx] \\
& \quad \quad \quad \quad \quad \quad \wedge pc' = \text{"Seek"} \\
& \quad \quad \wedge \text{UNCHANGED } \langle input, key, root, search, result, prefixCmp, stack \rangle \\
Result & \triangleq \wedge pc = \text{"Result"} \\
& \wedge result' = Iterate(iterStack) \\
& \wedge pc' = \text{"CheckResult"} \\
& \wedge \text{UNCHANGED } \langle iterStack, input, key, root, node, search, prefixCmp, \\
& \quad stack \rangle \\
CheckResult & \triangleq \wedge pc = \text{"CheckResult"} \\
& \wedge Assert(result = Expected(input, key), \\
& \quad \quad \quad \text{"Failure of assertion at line 194, column 3."}) \\
& \wedge pc' = \text{"Done"} \\
& \wedge \text{UNCHANGED } \langle iterStack, input, key, root, node, search, \\
& \quad result, prefixCmp, stack \rangle \\
\text{Allow infinite stuttering to prevent deadlock on termination.} \\
Terminating & \triangleq pc = \text{"Done"} \wedge \text{UNCHANGED } vars \\
Next & \triangleq findMin \vee Begin \vee Seek \vee Search \vee Consume \vee NextEdge \vee Result \\
& \quad \vee CheckResult \\
& \quad \vee Terminating \\
Spec & \triangleq Init \wedge \Box[Next]_{vars} \\
Termination & \triangleq \Diamond(pc = \text{"Done"}) \\
\text{END TRANSLATION}
\end{aligned}$$

\ * Modification History
\ * Last modified Fri Jul 02 08:15:31 PDT 2021 by mitchellh

* Created *Thu Jul 01 10:43:00 PDT 2021* by *mitchellh*